# $Id: PKGBUILD 176889 2013-02-02 04:33:21Z bisson $
# Contributor: Darwin M. Bautista <djclue917@gmail.com>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=ttf-sazanami
pkgver=20040629
pkgrel=7
pkgdesc='Japanese fonts'
url='http://wiki.fdiary.net/font/?sazanami'
license=('custom:BSD')
arch=('any')
depends=('fontconfig' 'xorg-fonts-encodings')
source=("http://osdn.dl.sourceforge.jp/efont/10087/sazanami-${pkgver}.tar.bz2")
sha1sums=('a6e558f49de8ab9884d92327905ce45698ca5429')

install=install

package() {
	cd "${srcdir}/sazanami-${pkgver}"
	install -d "${pkgdir}"/usr/share/{fonts/TTF,licenses/"${pkgname}"}
	install *.ttf "${pkgdir}/usr/share/fonts/TTF"
	iconv -f EUC-JP -t UTF-8 README > "${pkgdir}/usr/share/licenses/${pkgname}/README"
}
